What is this?

This notebook is part of a collection of `pluto` notebooks on various topics discussed during the Time Domain Astrophysics course delivered by Stefano Covino at the [Università dell'Insubria](https://www.uninsubria.eu) in Como (Italy). Please direct questions and suggestions to stefano.covino@inaf.it.

Table of Contents

Science Case: gamma-ray bursts

- GRBs are bright, rapid and short-duration gamma-ray signals appearing randomly in space and time. This is the light-curve of the first GRB ever detected, by one if the Vela satellites:



- After a few years to collect a sufficient number of events results were published and the GRB era began with this paper by Klebesadel et al. (1973).


THE ASTROPHYSICAL JOURNAL, 182:L85–L88, 1973 June 1
© 1973. The American Astronomical Society. All rights reserved. Printed in U.S.A.

OBSERVATIONS OF GAMMA-RAY BURSTS OF COSMIC ORIGIN

RAY W. KLEBESADEL, IAN B. STRONG, AND ROY A. OLSON

University of California, Los Alamos Scientific Laboratory, Los Alamos, New Mexico
Received 1973 March 16; revised 1973 April 2

- And since the very beginning the main question was to determine where do GRBs come from?
- A substantial, although indirect, breakthrough came after the launch of the Compton Gamma-Ray Observatory.



- Compton-GRO was Launched in 1991. One of the main goals of the mission was to study GRBs. In a few years of operation thousands of them were detected.

Extreme variability



- Rapid variability allows one to derive an estimate of the source size. It turns out to be very compact: of the order of tens of km.

The problem of GRB localization

- High-energy satellites can frequently detect GRBs thanks to wide-field of view capabilitied.
  - However, at the times, the localization capabilites yielded an uncertainty of the order of a degree radius.
  - In such an area typically you have billons of optical sources, Galactic and/or extra-Galactic.
- The problem can be addressed statistically.
  - First of all, the distribution of Galactic sources is far from isotropic in sky, as shown by the following all-sky view obtained by the Fermi telescope.



- On the contrary, the distribution of galaxies farther than ~ 100 Mpc is isotropic.
- Results based upon the observations of thousands of GRBs is impressive:



- The distribution is highly isotropic! GRBs form a population of cosmological sources.

The *BeppoSAX* contribution



- *Beppo*SAX was a Dutch-Italian satellite launched in 1996.
  - It had a revolutionary capability at the time: the ability to repoint after an alert in a few hours.
  - After that a GRB was located by the high-energy instrument with an error below a few tens of arcmin, the satellte repointed and the target location could be observed by small field of view but better resolution soft X-ray telescopes.
- This quickly brought to the identification of the first low-energy counterpart of a GRB: the afterglow era was beginning.



- Only a few monnths later, for GRB970508, the identification of the the optical counterpart was rapid enough to allow researchers to obtain a spectrum of the source that revealed to be hosted in a distant galaxy at a redshift $z \sim 0.8$.
- GRBs are now routinely observed at any redshift, from the closest case in the local universe to the faterthest objects ar redshift larger than 8.

- In the picture below an example of identification of the resdhift for GRB090726 is provided.



- The energy output of these events is huge, comparable to a SN but in a timescale of seconds!
- The cosmological origin of GRBs and their “extreme” features have been one of the hottest problems for astrophysics in the 2000s.

The need to be *Swift*

- The next step starts the golden age of GRB research, with the launch of the Neil Gehrels *Swift* observatory.



- Launched on 2004, Nov 20, and still operational.
- Swift has been designed to point very rapidly after an alert, typically with a time-scale of a minute.

So, what is a GRB?

- Let's summarize the results of decades of hard work in a plot:



- There are at least two families of GRBs: the long and short duration, with possible further subdivisions.



- We have observational evidence for both cases, although there could be more complex scenarios.

Exercise: the proposed QPO for GRB211211A

- GRB211211A was a bright GRB detected by several high-energy satellite that attracted a considerable attention since a ~ 20Hz QPOs have been reported in various phases of its prompt emission, e.g., Xiao et al. (2024) and Chirenti et al. (2024).
- This is a peculiarly difficult case: the GRB light-curve was short and the superposed oscillation could have lasted only a few cycles.
    - A non-stationary time-series with a transient behavior.
- Within this exercize we analyse Fermi-GBM data for this event.

- Data are stored in a FITS file. We extract the features of our interest and prepare the dataset to analyse.

```
# n0,n1,n2,n5,n9 and na, 8-200 keV
tte_file="glg_tte_na_bn211211549_v00.fit";
```

```
begin
    f = FITS(tte_file)

    ebound_hdu = f["EBOUNDS"]
    emin = read(ebound_hdu, "E_MIN")
    emax = read(ebound_hdu, "E_MAX")

    events_hdu = f["EVENTS"]
    time = read(events_hdu, "TIME")
    ch = read(events_hdu, "PHA")
end;
```

- Let's also define the energy range (keV):

```
begin
    E1 = 8.0
    E2 = 200.0

    index_for_EminEmax = (emin .>= E1) .& (emax .< E2)

    ch_arr = findall(index_for_EminEmax)

    ch1 = ch_arr[1]
    ch2 = ch_arr[end]

    ch_index = (ch .>= ch1) .& (ch .<= ch2)

    time_sel = time[ch_index]
    ch_sel = ch[ch_index]
end;
```

- And also the time window (seconds before and after the GRB time):

```
begin
    t1 = -0.5
    t2 = 1.0

    header_data = read_header(f[1])
    trigtime = header_data["TRIGTIME"]

    #trigtime = read_header(f[1])["TRIGTIME"]

    time_sel_sel = time_sel .- trigtime
    time_index = (time_sel_sel .>= t1) .& (time_sel_sel .<= t2)
    time_sel_T = time_sel_sel[time_index]
    ch_sel_T = ch_sel[time_index]
end;
```

- Finally, let's choose the binsize (0.01s):

```
begin
    binsize = 0.01

    tbins = t1:binsize:t2

    h = fit(Histogram, time_sel_T, tbins)

    histvalue = h.weights
    histbin = h.edges[1]
end;
```



- This episode of GRB activity is very short, and the pulse follows a typical profile with a verey fast rise and a slower, maybe exponential, decay.
- Let's now compute a LS periodogram for these data. A simple DFT could also be applied since the input light-curve is evenly sampled.

```julia
begin
    freq = range(1, 0.9/0.02, length=2000)

    final_t = convert(Vector{Float64}, vec(plottime))
    final_s = convert(Vector{Float64}, vec(plotrate))

    ls = lombscargle(final_t, final_s, frequencies=freq, normalization=:psd)

    pwrGBM = power(ls)
end;
```



- We see a peak at $\sim$ 20Hz, but we also see that the periodogram is characterized by a red-noise behavior with power quickly growing toward the lowest frequencies.
- This kind of behavior is rather typical, and it is possible to model the periodogram in order to test the significance of the detected period as we did, e.g., for the science case devoted to AGN and blazars (notebook, html).
- However, if we recall that a LS (or DFT) periodogram can be interpreted as a measure of the goodness of fit with sinusoids with different frequencies, it is likely that the red-noise behavior is (mainly) due to the modulation of the light-curve that is temptativey fit with low frequency sinusoids.
- We can try a different approach, i.e. fitting the pulse and removing it from the data and then analyze a “whitened” version of the light-curve.
    - Please, pay attention that in temporal analysis there are no free lunches. Removing the results of a fit implies that the residuals now suffer also from the accumulated uncertainties of the fit too. Resulting data are noisier.
    - In addition, unavoidably, after the fit removal, data are no more (if they were) independent of each other since the operation introduces some correlation.

Quite beyond the purpose of this exercize, let’s stress that any analysis of non-stationary time-series involves several additional complexities that must be properly evaluated. See, for instance, Hübner et al. (2022).

- In spite of these difficulties, we now model the pulse and remove it from the data and analyse the results.
- In order to model the pulse, we adopt a formula defined in Norris et al. (1996):

```
function Norris_1996_fast(x, A, tmax, sigma_r, sigma_d, mu, B)
    @. ifelse(x < tmax,
        B + A * exp(-(abs(x - tmax) / sigma_r)^mu),
        B + A * exp(-(abs(x - tmax) / sigma_d)^mu)
    )
end;
```

```
function norris_model(x, p)
    A, tmax, sigma_r, sigma_d, mu, B = p

    sr = abs(sigma_r)
    sd = abs(sigma_d)
    m  = abs(mu)

    return @. ifelse(x < tmax,
        B + A * exp(-(abs(x - tmax) / sr)^m),
        B + A * exp(-(abs(x - tmax) / sd)^m)
    )
end;
```

- And carry out a fit with this function:

```
begin
    p0 = [30.0, -0.01, 0.06, 0.3, 0.087, 1.0]

    lb = [-Inf, -Inf, 1e-6, 1e-6, 1e-6, -Inf]
    ub = [Inf, Inf, Inf, Inf, Inf, Inf]

    fit_result = curve_fit(norris_model, plottime, plotrate, p0, lower=lb, upper=ub)

    popt0GBM = fit_result.param

    pcov0GBM = estimate_covar(fit_result)

    errors = stderror(fit_result)
end;
```



- The fit looks like satisfactory, and therefore we now analyze the subtracted curve.
- Let's than compute a LS periodogram and also conpute FAP levels by means of a boostrapping techniques.

Moving block bootstrap

- Bootstrap is a powerful techniques that, of course, relies on a few assumptions.
- For *independent and identically distributed* (IID) bootstrap the data points are randomly sampled with replacement.
- For time series, the IID bootstrap destroys not only the periodicity but also any time correlation or structure in the time series. This might often results in underestimation of the confidence bars.
- For data with serial correlations it is better to use moving block (MB) bootstrap. In MB bootstrap blocks of data of a given length are patched together to create a new time series. The block length is a parameter. The ideal is to set the length so that it destroys the periodicity and preserves most of the serial correlation
- Bootstrap applied to time series is discussed in Bühlmann (2002) - "Bootstraps for time series.".

- Here, however, for didactic purposes, we first follow a simple approach with a standard bootstrap technique, later insetad we apply a MB.

```
begin
    ls_sub = lombscargle(plottime, Float64.(vec(sub_GBM)), frequencies=freq,
    normalization=:psd)

    pwrGBM_sub = power(ls_sub)
end;
```

```
function bootstrap_fap(times, signal, freq, probs; n_iterations=1000)
    max_powers = zeros(n_iterations)

    for i in 1:n_iterations
        shuffled_signal = shuffle(signal)

        ls_boot = lombscargle(times, shuffled_signal, frequencies=freq,
normalization=:psd)
        max_powers[i] = maximum(power(ls_boot))
    end

    return [quantile(max_powers, 1 - p) for p in probs]
end;
```

```
begin
    probabilities = [0.1, 0.05]

    resprGBM_boot = bootstrap_fap(plottime, Float64.(vec(sub_GBM)), freq,
    probabilities, n_iterations=1000)

end;
```

FAP levels: 90% (114.9), 95% (126.2)



- Given the assumptions, and of definite interest, the periodicity at ~ 20Hz does not show a strong significance.
- However, again, please consider that this is only an exemplificatory analysis, and the considerations discussed in Hübner et al. (2022) definitely hold.

- We have mentioned that red noise bahavior visible in the LS periodogram is likely due to the attempt of the algorithm to model the envelope of the light-curve evolution.
- We might wonder what the periodogram could become if we adopt a non-parametric method, e.g., the QMI, where the algorithm does not involve any modeling of the data.
- Coverting a light-curve from the time-domain to the phase-domain, given a period, typically destroyes the correlation affecting the data.

```
begin
    pg_qmi = Periodogram("qmieu")
    ffmin = 5
    ffmax = 1/0.02
    ffres = 1e-1
    fit!(pg_qmi, plottime, plotrate; dy = plotrate/10, fmin = ffmin, fmax = ffmax,
    resolution = ffres, n_local_optima=0);
end;
```



- We now see that the periodogram looks flat and the main peak is at $\sim 20Hz$ and higher harmonics.
- The main peak is not, anyway, truly dominating the periodogram. This sugegsts that it should not be highly significant.

- Standard bootstrapping assumes every data point is independent. In GRB lightcurves or power-law noise scenarios, points close in time are often correlated.
    - By picking a "block" of length block_length and moving it to the new time series, you preserve those correlations, making your significance tests (like FAP) much more robust against "fake" peaks caused by colored noise.
- We use a moving block bootstrap (or "overlapping block bootstrap") algorithm.
    - It's particularly useful for time-series data where the noise might be correlated over short timescales, as it preserves the local structure of the data better than a simple shuffle.

```julia
function block_bootstrap(mjd, mag, err; block_length=10.0, rseed=nothing)
    !isnothing(rseed) && Random.seed!(rseed)

    N = length(mjd)
    mjd_boot = zeros(Float64, N)
    mag_boot = zeros(Float64, N)
    err_boot = zeros(Float64, N)

    k = 1
    last_time = 0.0

    max_idx = 2
    for i in 2:N
        max_idx = i
        if mjd[end] - mjd[end - i + 1] > block_length
            break
        end
    end

    while k <= N
        idx_start = rand(1:(N - max_idx))

        idx_end = idx_start + 1
        for j in (idx_start + 1):N
            idx_end = j
            if mjd[idx_end] - mjd[idx_start] > block_length || k + (idx_end -
idx_start) >= N
                break
            end
        end

        len = idx_end - idx_start
        range_target = k:(k + len - 1)
        range_source = idx_start:(idx_end - 1)

        mjd_boot[range_target] .= mjd[range_source] .- mjd[idx_start] .+ last_time
        mag_boot[range_target] .= mag[range_source]
        err_boot[range_target] .= err[range_source]

        last_time = (mjd[idx_end] - mjd[idx_start]) + last_time
        k += len
    end

    return mjd_boot, mag_boot, err_boot
end;
```

```julia
begin
    niter = 200
    nmaxima = 20
    pbest_bootstrap = zeros(Float64, niter, nmaxima)
end;
```

- The QMI algorithm, we are applying here, is substabtially slower than the LS, and generating many artificial (boostrapped) light curves is a demanding task.

```
@progress for i in 1:niter
    tt_b, cc_b, ee_b = block_bootstrap(plottime, plotrate, plotrate/10,
block_length=0.3)
    pg_qmib = Periodogram("qmieu")
    fit!(pg_qmib, plottime, plotrate; dy = plotrate/10, fmin = ffmin, fmax = ffmax,
resolution = ffres, n_local_optima=0)
    pbest_bootstrap[i, :] = reverse(sort(pg_qmib.scores))[1:nmaxima]
end
```

100%

- In general, if one needs determine at which percentile in a simulated distrbution the oberved valiue can be located, it is necessary to have generated a sufficiently large number of simulated entries.
- The minimum number of simulated points depends on the accuracy one needs to obtain inm estimating the percentile.
- The error in a percentile estimate is roughly proportional to $1/\sqrt{N}$, where N is the number of simulations. However, percentiles are harder to estimate than means because they depend on the density of data at that specific point in the distribution.Since the 90th percentile is toward the “tail,” you have fewer data points in that region than you do at the median (50th percentile). Therefore, you need more simulations to achieve the same level of confidence.
- It is difficult to provide analytical solutions to the problem of determining the correct number of neeeded simulations. Probably, the best approach would be to choose of qa modest starting number, N, run the simulation a few times and check the concistency, within the requiremtns, of the results. If results are not satisfactory, increae the number of simulation by, say, an order ot magnitude and repeat.

Generalized Extreme value distribution

- In order to avoid the need of many iterations, it is possible to rely on the features of the Generalized Extreme value distribution.
- The extreme value theory, i.e. the attempt to evaluate the probability of occurrence of extreme measurements, is a rapidly evolving field of modern statistics. It is applied to many scenarios such as climatology, engineering, physics, economy and astronomy.
- The most interesting aspect of the GEV distribution is that it can be proved that it holds asymptotically, essentially independently of the probability distribution describing the phenomena that generate the observed data.
- There is some conceptual analogy with the central limit theorem (notebook, html) that guarantees that, under rather general conditions, the limiting distribution of sample means for a large family of parental statistical distributions is a standard normal distribution.
- The idea, essentially, is to carry out a simulation, fit the GEV (assuming, of course, that the kind of simulation, allows) and then extrapolate to the desidered percentile value.
- Example of applocation of the GEV in astronomy are, e.g., Süveges (2014) or Covino (2025).

▶ **More information about the GEV distribution?**

- So, we now fit a GEV distribution to the simulation results:

```julia
begin
    data = vec(pbest_bootstrap)


    function log_likelihood(p)
        μ, σ, ξ = p
        if σ <= 0
            return Inf
        end

        d = GeneralizedExtremeValue(μ, σ, ξ)

        return -loglikelihood(d, data)
    end

    p00 = [mean(data), std(data), 0.1]

    result = optimize(log_likelihood, p00)

    μ_best, σ_best, ξ_best = Optim.minimizer(result)
    d_gev = GeneralizedExtremeValue(μ_best, σ_best, ξ_best)


    quantile_val95 = quantile(d_gev, 0.95) # 95th percentile
    quantile_val90 = quantile(d_gev, 0.90) # 90th percentile

end;
```

90% quantile: 0.0008

95% quantile; 0.0009



Reference & Material

Material and papers related to the topics discussed in this lecture.

- Piran (2004) - “The physics of gamma-ray bursts”
- Hübner et al. (2022) - “Pitfalls of Periodograms: The Nonstationarity Bias in the Analysis of Quasiperiodic Oscillations”
- Huppenkothen et al. (2025) - “Searching for quasi-periodicities in short transients: the curious case of GRB 230307A”

Further Material

Papers for examining more closely some of the discussed topics.

- Klebesadel et al. (1973) - “Observations of Gamma-Ray Bursts of Cosmic Origin”
- Xiao et al. (2024) - “The Peculiar Precursor of a Gamma-Ray Burst from a Binary Merger Involving a Magnetar”
- Chirenti et al. (2024) - “Evidence of a Strong 19.5 Hz Flux Oscillation in Swift BAT and Fermi GBM Gamma-Ray Data from GRB 211211A”
- Norris et al. (1996) - “Attributes of Pulses in Long Bright Gamma-Ray Bursts”

Course Flow

	Previous lecture	Next lecture	Course Summary
notebook	Lecture about non-parametric periodograms	Lecture about singular spectrum analysis"	Course Summary
html	Lecture about non-parametric periodograms	Lecture about singular spectrum analysis	Course Summary

Notebook v1.0.0 - 06 May 2026